
付録 B Prolog超簡単入門

Prologは、論理型言語と呼ばれるプログラミング言語の仲間のうち、もっとも代表的なものであ
る。論理型言語では、「～ならば～」という論理式の集まりをプログラムとみなし、論理式の証明を
プログラムの実行とみなす。

Prologなどの論理型言語に特徴的な概念としては、ユニフィケーション（単一化）、バック
トラッキング（後戻り）、論理変数などが挙げられる。

B.1 Prologでのプログラミング

Prologでは「～ならば～」という論理式の集まりをプログラムとみなすが、この「～」の部分を構
成するのが、次のような形式である。

述語 (項 1, . . . , 項 n)

このかたちを素論理式1と呼ぶ。「項」は数・文字列などの定数や変数、あるいはリストなどのデー
タ構造などである。「述語」は直観的には項 1, . . . ,項 nの間の関係を表す。例えば、father(adam,

cain)という素論理式は「Adamは Cainの父である」という関係を表す。
Prologのプログラムは、ホーン節（Horn clause）と呼ばれる形式の集まりである。ホーン節とは、

素論理式 0 :- 素論理式 1, . . . , 素論理式 n.

という形である。（最後に必ずピリオド（.）が必要である。）:-は左向きの矢印（←）と考えれば良
い。つまり、これは素論理式 1, . . . ,素論理式 nがすべて成り立つならば、素論理式 0も成り立つとい
う規則を表している。
ホーン節の:-の左側には素論理式は一つのみであり、これをヘッド（head）と呼ぶ、:-の右側の

素論理式の並びはボディ（body）と呼ぶ。
例えば、

grandchild(X, Z) :- child(X, Y), child(Y, Z).

は、Xと Y, Yと Zの間に childという関係がある（Xが Yの子であり、Yが Zの子である）ならば、
Xと Zの間に grandchildという関係がある（Xは Zの孫である）という規則を述べている。ここで
X, Y, Zは変数である。Prologの変数は X, Xs, Yのように、アルファベットの大文字（あるいはアン
ダースコア “_”）からはじまる識別子（名前）を使用する。一方、小文字からはじまる識別子は、述
語や構成子などの定数に利用される2。変数はどのような項に具体化しても良いので、変数を使用し
た規則は、実際には無数の規則を表していることになる。

1アトム（Atom）、複合項などと呼ぶこともある。
2Haskellと全く逆なので注意する。

ホーン節のうちボディがないものを事実（ fact）と言う。このときは、:-も省略する。（文末のピ
リオドは必要である。）例えば、

child(hidetada, ieyasu).

は hidetada（秀忠）が ieyasu（家康）の子である、という事実を表明している。ここで hidetada,

ieyasuは小文字からはじまるので定数である。
Prologのプログラムは、このようなホーン節（事実を含む）を集めたものである。例えば徳川家の

家系図を表すプログラムの一部は次のようになる。

grandchild(X, Z) :- child(X, Y), child(Y, Z).

child(hideyasu, ieyasu).
child(hidetada, ieyasu).
child(yoshinao, ieyasu).
child(yorinobu, ieyasu).
child(yorifusa, ieyasu).

child(iemitsu, hidetada).
child(tadanaga, hidetada).
child(masayuki, hidetada).

child(ietsuna, iemitsu).
child(tsunayoshi, iemitsu).

このようなプログラムをファイルに作成し、処理系にロード 3する。例えば Windows上で動作する
SWI-Prologという処理系ではメニューの「File」—「Consult. . . 」でプログラムファイルをロードす
ることができる。
プログラムは、このようなホーン節の集まりに対して質問を発することにより起動される。Prolog

の処理系は「?- 」というプロンプトを出力するので、このあとに質問を入力する。（質問も最後に必
ずピリオドが必要である。）

?- grandchild(hidetada, ieyasu).

No
?- grandchild(iemitsu, ieyasu).

Yes

1番目の質問は、Hidetada（秀忠）は Ieyasu（家康）の孫か?という質問である。これはプログラム中
の規則から導出できないので、処理系は Noと答えている。2番目の質問は、Iemitsu（家光）は家康の
孫か? という質問である。これは、 child(hidetada, ieyasu).と child(iemitsu, hidetada).

という二つの事実と

grandchild(X, Z) :- child(X, Y), child(Y, Z).

というホーン節から導出できるので Yesと答えている。
さらに Prologでは質問の中に変数を含めることができる。すると Prologの処理系は、その質問
を成り立たせる変数への代入を出力する。例えば、

?- grandchild(X, ieyasu).

3Prologでは伝統的に consultという。

という質問に対しては、

X = iemitsu

という解を出力する。ここで、リターンキーを押すとこれで終ってしまうが、「;」を入力すると、さ
らに別解を表示させることができる。

X = iemitsu ;

X = tadanaga ;

X = masayuki ;

No

質問には一般に複数の素論理式を並べることができる。

?- 素論理式 1, . . . , 素論理式 n.

このような形をゴール節と呼ぶ。直観的には、並べた素論理式がすべて成り立つか?（成り立たせ
るような変数への代入が存在するか?）という質問を表している。

B.2 単一化（ユニフィケーション）と後戻り

この節では、Prologのプログラムの実行方法を解説する。解説を簡単にするために、まず、最初に
いくつかの言葉を定義しておく。

2つ以上の（通常変数を含む）素論理式があり、変数に適切な代入をして、これらの素論理式をまっ
たく同一のものにすることができるとき、これらの素論理式は単一化可能（unifiable）であると

いう。また、その時の代入を単一化代入（unifier）という。例えば、次の 2つの素論理式

p(a, Y, Z) と p(X, b, Z)

は X = a, Y = bを単一化代入として単一化可能である。単一化可能な時は、通常必要なのは最汎
（最も一般的な）単一化代入（most general unifier,MGU）である。例えば、上の例では、X = a, Y

= b, Z = cという代入も単一化代入であるが、前者の方がより一般的である。実際、この例の場合
は X = a, Y = bが MGUになる。
プログラム中のホーン節

P :- Q1, Q2,
. . . , Qn,

のヘッド Pが素論理式（G）と単一化可能なとき、このホーン節は Gに適用できるという。
Prologのプログラムの実行方法は、次のようにまとめられる。

1. ゴール節中のもっとも左の素論理式に対し、適用できるホーン節を選ぶ。適用できるホー
ン節が複数ある時は、プログラム中に先に書かれているホーン節から試みる4。

4Prologを定理証明系として見た時には、ここで、「最も左」、「先に書かれている」という選択をするために、証明系と
しての力が弱くなってしまう。つまり、このような制限をしなければ証明できるはずの論理式が証明できなくなってしま
う。しかし、プログラミング言語として見た時は、効率を確保するために必要な制限である。

2. 選んだ適用可能なホーン節に対して、そのMGUをゴール節の残りの素論理式に適用し、さら
にゴール節中の最左素論理式を、上で選んだホーン節のボディに MGUを適用したものと置き
換える。（ホーン節のボディがないときは、最左素論理式を消す。）

3. もし適用できるホーン節がなければ、後戻り（バックトラック）して、ひとつの前のゴー
ル節中の素論理式がある状態からやり直す。そして、その素論理式に適用できるホーン節のう
ち次の候補を選ぶ。

4. ゴール節に素論理式がなくなれば、プログラムの実行を終了し、その時得られた変数への代入
を表示する。素論理式がまだあれば、1に戻る。

具体的に、

?- grandchild(X, ieyasu).

というゴール節での動作を説明することにする。まずこのゴール節の最左（1つしかないが）素論理
式は grandchild(X, ieyasu)である。これに適用できるホーン節は、今のプログラムでは、

grandchild(X, Z) :- child(X, Y), child(Y, Z).

しかなく、最汎単一化代入は Z = ieyasuである。するとゴール節は

?- child(X, Y), child(Y, ieyasu). · · · 1©

に変換される。
次に、この最左の child(X, Y)に適用できる最初のホーン節は、

child(hideyasu, ieyasu).

であり、MGUは X = hideyasu, Y = ieyasuである。すると、ゴール節は、

?- child(ieyasu, ieyasu).

に変換される。しかし、このゴール節に適用できるホーン節はない。
そこで、上の 1©のところまでバックトラックし、次の候補である
child(hidetada, ieyasu).

の適用を試みる。しばらくのあいだ同様にバックトラックが続き、最終的に 1©のゴール節に対し、
child(iemitsu, hidetada).

というホーン節を選ぶ。そのとき MGUは X = iemitsu, Y = hidetadaとなる。すると、ゴール
節は、

?- child(hidetada, ieyasu).

に書き換えられる。これはすぐに適用できるホーン節が見つかり、ゴール節が空になって、結果とし
て代入:

X = iemitsu

が出力される。ここで;が入力されると、またバックトラックが起こる。
ホーン節の適用は命令型言語・関数型言語の関数呼出しのようなものだと考えることもできるが、
単一化により呼び出される側（ホーン節のヘッド）から呼び出す側（ゴール節）に情報が流れること
がある、というところが論理型言語に特徴的なところである。例えば、

?- child(X, ieyasu).

という呼出しに対しては、X = hideyasu（あるいは hidetada, yoshinao . . . ）という情報が、呼び
出された側から、呼び出した側に伝えられる。

Prologの変数は命令型言語の変数と異なり、いったん単一化により値が代入されれば、以降は値を
変更されることはない。ただし、（純）関数型言語の変数とも異なり、最初はいったん不定（unknown）
な値を取ることができる。このような振舞いをする変数は一般に論理変数（ logical variable）と呼
ばれる。

B.3 Prologでのリスト処理

Prologでのリストの記法は要素を,で区切り、角括弧（[と]）で囲んで、[1, 2, 3]のように書く。
また、先頭の要素が X で、先頭を除く残りの要素からなるリストが Xs であるようなリストは

[X|Xs]と表記される。これは Haskellの x:xs という書き方に相当する。また、[1, 2, 3] は

[1|[2|[3|[]]]]の略記法である。
Prologでのリストを連接（append）するプログラムは次のように記述できる。

append([], Y, Y).
append([H|X], Y, [H|Z]) :- append(X, Y, Z).

これは、1番目の引数と 2番目の引数を連接した結果が 3番目の引数になる、という関係を表して
いる。
例えば, [1, 2]と [3, 4]の連接は次のように求められる。

?- append([1, 2], [3, 4], X).

X = [1, 2, 3, 4] ;

No

問 B.3.1 appendの逆向きの計算ができることを、前節で示した Prologの実行方法で実際に 1ステッ
プずつ確認せよ。

上の問を解いてみるとわかるが、変数 Xは後ろの要素が不定（変数）のまま、前の要素から順に定
まっていく。後ろの要素がもっと後で定まるようなプログラムを書くことも可能である。例えば、

myconcat([], []).
myconcat([Xs|Xss], Ys) :- append(Xs, Zs, Ys), myconcat(Xss, Zs).

これは、リストのリストを平坦なリストに変換するプログラムである。
?- myconcat([[1, 2], [3, 4], [5, 6]], Xs).

Xs = [1, 2, 3, 4, 5, 6] ;

No

appendの第 2引数の Zsはあとの myconcatの呼出しで値が定まる。appendの第 3引数の Ysはし
ばらくの間、最後が論理変数で終る形で扱われることになる。
このように cdr部分が論理変数になっているリストを開リスト（open list）という。開リストは

Prologでリストを扱う時に良く使われるイディオムである。
さらに Prologのおもしろいところは appendの逆向きの計算もできるということである。
?- append(X, Y, [1, 2, 3]).

X = []
Y = [1, 2, 3] ;

X = [1]
Y = [2, 3] ;

X = [1, 2]
Y = [3] ;

X = [1, 2, 3]
Y = [] ;

No

この例では連接して [1, 2, 3]になる 2つのリストの、すべての可能性を求めていることになる。
ユーザが「;」を入力するたびにバックトラックが起こり別解を表示する。

問 B.3.2 実際に、上のような結果が出ることを、上に示した Prologの実行方法で 1ステップずつ確
認せよ。

